

Chapter 4

Datatypes

The types we have seen so far, that is functions ($_ \rightarrow _$), products ($_ \times _, \top$) and sums ($_ + _, \perp$) are finite or preserve finiteness. In computation but also in Mathematics we often use infinite types, such as the natural numbers (\mathbb{N}), which I have already used for illustration. In this chapter we will look at such *datatypes*, starting with the natural numbers as the canonical example. We discuss primitive recursion first in a schematic way and then given by combinators: the recursor which can be reduced to the iterator (also called fold). We then look at various examples of inductive types such as lists, syntax trees and infinite trees representing ordinal notations. We investigate the limits of what constitutes a reasonable inductive type and what not — this is related to the notion of positivity. Using the notion of categorical mirror we also investigate the dual of inductive types: coinductive types. Examples here are streams and coinductive natural numbers, which can also be infinite.

4.1 The natural numbers

One of the most versatile datatypes we consider is the type of natural numbers, i.e. the counting numbers $0, 1, 2, 3, 4, \dots$. Indeed, because we are computer scientists we are starting counting with 0, which is useful when calculating off-sets in datastructures but it is also the answer to the question *How many elephants do you have in your fridge* (I hope).

It was Guisepe Peano from Torino who wrote down the precise laws of the natural numbers in predicate logic at the end of the 19th century, this is nowadays known as Peano Arithmetic. We aren't going to study these laws in detail but use one important idea: according to Peano a natural number is either zero or it is the successor of another natural number. We can implement this in Agda:

```
data  $\mathbb{N}$  : Set where  
  zero :  $\mathbb{N}$   
  suc  :  $\mathbb{N} \rightarrow \mathbb{N}$ 
```

This is the definition of a sum, but the novelty is that it is recursive, that is we are using \mathbb{N} which is just being defined in the argument of the constructor `suc`.

Given this we can derive the usual notation: ¹

$$\begin{array}{ll} 0 = \text{zero} & \\ 1 = \text{suc } 0 & = \text{suc zero} \\ 2 = \text{suc } 1 & = \text{suc (suc zero)} \\ 3 = \text{suc } 2 & = \text{suc (suc (suc zero))} \\ \vdots & \vdots \end{array}$$

Let's define a function on the natural numbers, for example the predecessor:

```
pred : ℕ → ℕ
pred zero   = zero
pred (suc n) = n
```

Here we use pattern matching to analyse the constructor as we have seen already. Actually this function is a bit of a cheat because there isn't really a predecessor of 0. A better version uses `Maybe` where `Maybe A` is just a shorthand for $\top \uplus A$:

```
data Maybe (A : Set) : Set where
  nothing : Maybe A
  just    : A → Maybe A
```

Using `Maybe` we can now define a version of `pred` that signals an error on 0 by returning `nothing`:

```
pred : ℕ → Maybe ℕ
pred zero   = nothing
pred (suc n) = just n
```

This version of `pred` indicates an error on 0 by returning `nothing` but for any other number returns `just` the predecessor.

This predecessor is the exact inverse of the constructors which can be combined into one function:

```
zerosuc : Maybe ℕ → ℕ
zerosuc nothing = zero
zerosuc (just n) = suc n
```

It is not hard to see that the functions `pred` and `zerosuc` are inverse to each other.

¹Luckily we can tell Agda to use the standard notation for numbers simply by invoking a *pragma*:

```
{-# BUILTIN NATURAL ℕ #-}
```

4.1.1 Recursion

To define more interesting functions on the natural numbers we need to use recursion, for example let's define the doubling function:

$$\begin{aligned} \text{double} &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{double zero} &= \text{zero} \\ \text{double (suc } n) &= \text{suc (suc (double } n)) \end{aligned}$$

That is the double of 0 is 0 and the double of a number of the form $\text{suc } n$ is $2 + \text{double } n$, which as we haven't yet defined $_ + _$ is just $\text{suc (suc (double } n))$.

For example $\text{double } 3 = 6$. Let's go through this step by step:

$$\begin{aligned} \text{double } 3 &= \text{double (suc (suc (suc zero)))} \\ &= \text{suc (suc (double (suc (suc zero))))} \\ &= \text{suc (suc (suc (suc (double (suc zero)))))} \\ &= \text{suc (suc (suc (suc (suc (suc (double zero))))))} \\ &= \text{suc (suc (suc (suc (suc zero))))} \\ &= 6 \end{aligned}$$

We can justify the use of recursion: all natural numbers are constructed via zero and suc . To define a function f over the natural numbers we just have to give a case for zero and one for $\text{suc } n$. Since the number n has to be constructed *before* we $\text{suc } n$ we can construct the answer to the function f in the same way and compute the answer to $f n$ before we compute the answer to $f (\text{suc } n)$ hence we can use $f n$ when constructing $f (\text{suc } n)$.

Not all functions are exactly of this form, already the inverse of double , that is half which forgets the remainder, uses a slightly more general scheme

$$\begin{aligned} \text{half} &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{half zero} &= \text{zero} \\ \text{half (suc zero)} &= \text{zero} \\ \text{half (suc (suc } n)) &= \text{suc (half } n) \end{aligned}$$

That is half of 0 and 1 is zero and half of $2 + n$ is $1 + \text{half } n$. The recurrence from $\text{suc (suc } n)$ can be justified in the same way as before because certainly n is constructed before $\text{suc (suc } n)$.

4.1.2 Arithmetic operations

Let's define the standard functions of arithmetic starting with addition:

$$\begin{aligned} _ + _ &: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{zero} + n &= n \\ \text{suc } m + n &= \text{suc (} m + n) \end{aligned}$$

The idea of this definition of addition is that for example $3 + 5$ is the 3rd successor of 5 that $\text{suc}(\text{suc}(\text{suc } 5))$. This means that $0 + n$ is just n and $\text{suc } m + n$ is one more than $m + n$.

Next we define multiplication $_ * _$ which is repeated addition in the same way as $_ + _$ is repeated suc .

$$\begin{aligned} _ * _ &: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{zero} * n &= 0 \\ \text{suc } m * n &= n + m * n \end{aligned}$$

That is for example $3 * 5$ is $5 + 5 + 5 + 0$. Recursively we express this by saying that $0 * n$ is just 0 and $\text{suc } n * m$ is $n + (m * n)$.

We can go further and define exponentiation as repeated multiplication. In the previous cases it didn't matter over which argument we recursed because they were commutative. But exponentiation certainly isn't, e.g. $2^3 = 8$ but $3^2 = 9$. If we want to mimic the usual order of arguments for exponentiation we need to recur on the 2nd argument:

$$\begin{aligned} _ \uparrow _ &: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ m \uparrow \text{zero} &= 1 \\ m \uparrow \text{suc } n &= m * m \uparrow n \end{aligned}$$

For example $3 \uparrow 5$ is $3 * 3 * 3 * 3 * 3 * 1$. Hence $m \uparrow 0 = 1$ and $m \uparrow \text{suc } n = m * m \uparrow n$.

4.1.3 The Ackermann function

Each instance of recursion correspond to using a for-loop in conventional programming. While addition is just defined using one for-loop, multiplication calls addition inside its loop, hence we have a nested for-loop. And exponentiation uses a double nested for-loop. Historically functions that only use for-loops are called primitive recursive, and all the examples we have seen are examples of primitive recursion. Hilbert, a famous german Mathematician, asked wether all total recursive functions can be defined using primitive recursion. His student Ackermann showed that this is not the case by exhibiting a function that is not primitive recursive because it grows faster than any primitive recursive function. The idea of the Ackermann function is to have one extra parameter which determines the nesting of for loops, that is he defined a function $\text{ack} : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$ such that

$$\begin{aligned} \text{ack } 0 \ m \ n &= \text{suc } n \\ \text{ack } 1 \ m \ n &= m + n \\ \text{ack } 2 \ m \ n &= m * n \\ \text{ack } 3 \ m \ n &= m \uparrow n \\ &\vdots \end{aligned}$$

The next function, hyperexponentiation that is repeated exponentiation doesn't really have a standard notation anymore. We can define `ack` using recursion over the natural numbers:

```

ack : ℕ → ℕ → ℕ → ℕ
ack zero      m n      = suc n
ack (suc zero) m zero  = m
ack (suc (suc zero)) m zero = 0
ack (suc (suc (suc zero))) m zero = 1
ack (suc (suc (suc (suc l)))) m zero = m
ack (suc l)   m (suc n) = ack l m (ack (suc l) m n)

```

To stay consistent with exponentiation we define the functions by recursion over the 2nd argument. The first line defines `ack 0` to be just the successor of the 2nd argument because then we can obtain addition by repeating it. The next three lines define the 0 cases for addition $m + 0 = m$, multiplication $m * 0 = 0$ and exponentiation $m \uparrow 0 = 1$. The idea for multiplication and exponentiation is to use the neutral element of the previous operation, corresponding to 0 iterations. However, since there is no neutral element for exponentiation, that is no number such that $x^n = n$, we just use the first argument as for addition. The last line uniformly expresses that the function for `ack (suc l)` is defined by iterating the previous function `ack l n` times.

Actually the definition above is a bit complicated due to all the different cases for 0, hence often a simplified version of the Ackermann function is used where the levels don't exactly correspond to the usual definitions of the arithmetic operations but have the same nesting of for loops:

```

acks : ℕ → ℕ → ℕ → ℕ
acks zero  m n      = suc n
acks (suc l) m zero  = m
acks (suc l) m (suc n) = acks l m (acks (suc l) m n)

```

Using either `ack` or `acks` we can define the function `fast`:

```

fast : ℕ → ℕ
fast n = ack n n n

```

We can only evaluate the first few instances of `fast`

```

fast 1 = 1 + 1           = 2
fast 2 = 2 * 2           = 4
fast 3 = 3 ↑ 3           = 27
fast 4 = 4 ↑ (4 ↑ (4 ↑ 4)) = ?

```

Due to the fact that our implementation of numbers using a unary representation we can't even compute `fast 4` in a reasonable time. But even using a very efficient implementation of numbers will not save us much, the incredible `fast`

growing behaviour of `fast` will spoil any attempt to compute many more results. And indeed, `fast` grows faster than any primitive recursive function that is any function definable only with for-loops.

4.2 Iterator and recursor

We intuitively explained and justified recursive definitions over the natural numbers but now we shall be more precise and say exactly what recursive definitions are permissible.

In general we define a function $f : \mathbb{N} \rightarrow M$ ² by recursion over the natural numbers

$$\begin{aligned} f\ 0 &= z \\ f\ (\text{suc } n) &= s\ (f\ n) \end{aligned}$$

where

$$\begin{aligned} z &: M \\ s &: M \rightarrow M \end{aligned}$$

We can turn this into a higher order function, the iterator³:

$$\begin{aligned} \text{It}\mathbb{N} &: M \rightarrow (M \rightarrow M) \rightarrow \mathbb{N} \rightarrow M \\ \text{It}\mathbb{N}\ z\ s\ \text{zero} &= z \\ \text{It}\mathbb{N}\ z\ s\ (\text{suc } n) &= s\ (\text{It}\mathbb{N}\ z\ s\ n) \end{aligned}$$

⁴ We call M the *motive* and z,s methods. Our claim is that we can reduce recursion over the natural number to the iterator. Ok, we haven't actually made precise what we mean by recursion over the natural numbers and introducing the iterator is exactly one way to do this.

We can derive `dbl` using only the iterator:

$$\begin{aligned} \text{double-it} &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{double-it} &= \text{It}\mathbb{N}\ 0\ (\lambda\ \text{double-n} \rightarrow \text{suc } (\text{suc } \text{double-n})) \end{aligned}$$

The instances for the methods z and s can be read off the recursive definition of `double`:

$$\begin{aligned} \text{double} &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{double zero} &= \text{zero} \\ \text{double } (\text{suc } n) &= \text{suc } (\text{suc } (\text{double } n)) \end{aligned}$$

The reduction of `half` is less obvious - here again the recursive definition:

²In the following I use $A\ B\ M : \text{Set}$ as variables as in chapter 2.

³In category theory $\text{It}\mathbb{N}\ z\ s : \mathbb{N} \rightarrow M$ is the universal morphism from the initial algebra. It is also called a *fold* in functional programming, or a *catamorphism*.

⁴Remember that M is one of the names we have declared to be used as implicit parameters for `Set`.

```

half : ℕ → ℕ
half zero      = zero
half (suc zero) = zero
half (suc (suc n)) = suc (half n)

```

`half` doesn't follow exactly the pattern we have stated before but it is reducible to it. The idea is that we define an auxiliary function `half-aux` which computes the results of `half n` and `half (suc n)` simultaneously using a product.

```

half-aux : ℕ → ℕ × ℕ
half-aux zero = zero , zero
half-aux (suc n) = proj2 (half-aux n) , suc (proj1 (half-aux n))

```

⁵ The first line is justified by the observation that `half 0` , `half 1` = `0` , `0`. The recursive call follows from the fact that we have already `half (suc n)` as the 2nd component of `half-aux n` and by the recursive definition `half (suc (suc n))` = `half n` which explains the 2nd component. Now `half-aux` is an instance of our scheme and we can define

```

half-aux-it : ℕ → ℕ × ℕ
half-aux-it = ltℕ (zero , zero) (λ p → proj2 p , suc (proj1 p))
half-it : ℕ → ℕ
half-it n = proj1 (half-aux-it n)

```

The definition of `_+_` is straightforward, here the recursive version:

```

_+_ : ℕ → ℕ → ℕ
zero + n = n
suc m + n = suc (m + n)

```

We can transform it into an instance of our scheme using higher order functions:

```

zero +_ = λ n → n
suc m +_ = λ n → suc (m + n)

```

And hence we define

```

_+_it_ : ℕ → ℕ → ℕ
_+_it_ = ltℕ (λ n → n) (λ m+n → suc (m+n))

```

⁵The 2nd line can be more nicely written using a local definition:

```

half-aux (suc n) = proj2 hn , suc (proj1 hn)
  where hn = half-aux n

```

or even better using `with` which enables pattern matching on an intermediate result:

```

half-aux (suc n) with half-aux n
half-aux (suc n) | half-n , half-sn = half-sn , (suc half-n)

```

The same technique works for multiplication:

$$\begin{aligned} _!-it_ &: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ _!-it_ &= \text{lt}\mathbb{N} (\lambda n \rightarrow 0) \lambda m^* n \rightarrow n + (m^* n) \end{aligned}$$

A more interesting case is a function which doesn't only use the recursive result but also the input. An example is the factorial function which can be recursively defined we follows:

$$\begin{aligned} _! &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{zero } ! &= 1 \\ \text{suc } n ! &= \text{suc } n * n ! \end{aligned}$$

$n!$ is the number of ways we can put n people on n chairs. If there are no people and no chairs we have to do nothing hence the answer is 1 (this seems to confuse some people). Otherwise in the case $(\text{suc } n)!$ there are $\text{suc } n$ ways to place the first person and then $n!$ to distribute n people on n chairs.

The factorial function is an instance of a slightly modified scheme, to define $f : \mathbb{N} \rightarrow A$ we use:

$$\begin{aligned} f 0 &= z \\ f (\text{suc } n) &= s n (f n) \end{aligned}$$

where

$$\begin{aligned} z &: A \\ s &: \mathbb{N} \rightarrow A \rightarrow A \end{aligned}$$

As above we can turn this into a general higher order function, called the recursor ⁶:

$$\begin{aligned} \text{RN} &: M \rightarrow (\mathbb{N} \rightarrow M \rightarrow M) \rightarrow \mathbb{N} \rightarrow M \\ \text{RN } z s \text{ zero} &= z \\ \text{RN } z s (\text{suc } n) &= s n (\text{RN } z s n) \end{aligned}$$

We can define the factorial function using only the recursor:

$$\begin{aligned} _!-r &: \mathbb{N} \rightarrow \mathbb{N} \\ _!-r &= \text{RN } 1 (\lambda n n! \rightarrow \text{suc } n * n!) \end{aligned}$$

We can actually derive RN from $\text{lt}\mathbb{N}$ by (re-)computing the input and the output together:

$$\begin{aligned} \text{RN-it} &: M \rightarrow (\mathbb{N} \rightarrow M \rightarrow M) \rightarrow \mathbb{N} \rightarrow M \\ \text{RN-it } \{M\} z s n &= \text{proj}_2 (\text{lt}\mathbb{N} z' s' n) \\ \text{where} \\ z' &: \mathbb{N} \times M \end{aligned}$$

⁶Called *paramorphism* by Meertens.

$$\begin{aligned} z' &= 0, z \\ s' : \mathbb{N} \times \mathbb{M} &\rightarrow \mathbb{N} \times \mathbb{M} \\ s'(n, m) &= \text{suc } n, s \ n \ m \end{aligned}$$

That is we use $\mathbb{N} \times \mathbb{M}$ as the motive, the first component is identical to the input. Hence in the 0-case we use $0, z$ and in the successor case given n, m we compute $\text{suc } n, s \ n \ m$, hence we are able to supply the extra argument. In the end we have only to extract the 2nd component to obtain the result.

Finally let's have a look at the simplified Ackermann function:

$$\begin{aligned} \text{acks} : \mathbb{N} &\rightarrow \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{acks zero } m \ n &= \text{suc } n \\ \text{acks (suc } l) \ m \ \text{zero} &= m \\ \text{acks (suc } l) \ m \ (\text{suc } n) &= \text{acks } l \ m \ (\text{acks (suc } l) \ m \ n) \end{aligned}$$

Can we compute the Ackermann function using the iterator or the recursor? It seems that our scheme exactly captures primitive recursion. However, the traditional notion of primitive recursion doesn't include the use of higher order functions which leads to a much more powerful scheme. And indeed the Ackermann function is definable:

$$\begin{aligned} \text{acks-it} : \mathbb{N} &\rightarrow \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{acks-it} &= \text{ltN } (\lambda \ m \ n \rightarrow \text{suc } n) \\ &(\lambda \ \text{ack-l } m \rightarrow \text{ltN } m \ \lambda \ \text{ack-sl-m-n} \rightarrow \text{ack-l } m \ \text{ack-sl-m-n}) \end{aligned}$$

The main idea here is to translate the pattern using higher order functions as in the case for addition and multiplication, and then use the iterator again to translate the nested pattern.

Indeed, the power of primitive recursion with higher order function is much more than ordinary primitive recursion. We can define all recursive functions that can be proven total in Peano Arithmetic. Indeed a lambda calculus with a type of natural numbers and the recursor is called *System T* and was introduced by Gödel as a functional system corresponding to Peano Arithmetic.

4.2.1 Non primitive recursive functions

There are recursive functions over the natural numbers that cannot be defined directly using primitive recursion. An example is the computation for the *greatest common divisor* using subtraction. First we define *cut-off subtraction* which stops at 0:

$$\begin{aligned} _ - _ : \mathbb{N} &\rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{zero} - n &= \text{zero} \\ \text{suc } m - \text{zero} &= \text{suc } m \\ \text{suc } m - \text{suc } n &= m - n \end{aligned}$$

That is $3 - 2 = 1$ while $1 - 2 = 0$. Another ingredient is a comparison function *comp* which gets two numbers m, n and returns different values depending on whether $m < n$, $m = n$ or $m > n$:

```

comp : ℕ → ℕ → A → A → A → A
comp zero zero m<n m=n m>n      = m=n
comp zero (suc n) m<n m=n m>n    = m<n
comp (suc m) zero m<n m=n m>n    = m>n
comp (suc m) (suc n) m<n m=n m>n = comp m n m<n m=n m>n

```

Now we can define `gcd` where we always subtract the smaller number from the larger one until they are equal:

```

gcd : ℕ → ℕ → ℕ
gcd m n = comp m n (gcd m (n - m)) m (gcd (m - n) m)

```

e.g. `gcd 56 84 = 27`.

However, Agda complains *Termination checking failed* because it cannot see that the arguments were reduced during recursion.⁷ And indeed, there is no obvious way to translate this function into one using the iterator or the recursor.

In this case we can save the day by noticing that the recursion is bounded by the sum of the numbers and define:

```

gcd-aux : ℕ → ℕ → ℕ → ℕ
gcd-aux zero m n = zero
gcd-aux (suc fuel) m n =
  comp m n (gcd-aux fuel m (n - m)) m (gcd-aux fuel (m - n) m)
gcd' : ℕ → ℕ → ℕ
gcd' m n = gcd-aux (m + n) m n

```

The first argument is often called `fuel` because it acts like the fuel for a car, we have to pay one unit for each recursive call. However, we have set up our definition so that we can never run out of fuel. Hence the first line in the definition of `gcd-aux` will never be executed and we could have put in any value instead of 0. There are better ways to deal with this situation, especially once we have dependent types.

However, there are recursive and terminating functions which we cannot define using higher order primitive recursion.

4.3 Inductive Types

4.3.1 Lists

The type of natural numbers

```

data ℕ : Set where
  zero : ℕ
  suc  : ℕ → ℕ

```

⁷We can override this warning by using the pragma

```
{-# TERMINATING #-}
```

before the declaration of the function.

of an *inductive datatype* which we can view as a recursively defined sum. Another example of an *inductive datatype* is the type of lists, representing finite sequences of a given type:

```
data List (A : Set) : Set where
  [] : List A
  _::_ : A → List A → List A
```

Constructors in this case are the empty list [] (pronounced *nil*) and given $a : A$ and another list $l : \text{List } A$ we can construct a new list $a :: l$ using the constructor $_{::}$ ⁸ (pronounced *cons*). All lists are made from [] and $_{::}$, for example $1 :: 2 :: 3 :: [] : \text{List } \mathbb{N}$.

Recursive definitions on lists follow the same idea as for natural numbers. Examples are the length function:

```
length : List A → ℕ
length [] = 0
length (x :: l) = 1 + length l
```

which computes the length of a list; a function that sums the numbers in a list:

```
sumList : List ℕ → ℕ
sumList [] = 0
sumList (n :: ns) = n + sumList ns
```

and the append function which concatenates two lists:

```
_++_ : List A → List A → List A
[] ++ ys = ys
(x :: xs) ++ ys = x :: (xs ++ ys)
```

So for example

```
length (1 :: 2 :: 3 :: []) = 3
sumList (1 :: 2 :: 3 :: []) = 6
(1 :: 2 :: 3 :: []) ++ (4 :: 5 :: []) = 1 :: 2 :: 3 :: 4 :: 5 :: []
```

Append is very similar to addition, indeed addition can be viewed as a special case of append if we identify \mathbb{N} with $\text{List } \top$. However, unlike $_{++}$, $_{++}$ is not commutative in general, since the order of elements matter.

Lists are a functor, that is an operation that can not only applied to elements but also to functions. This is witnessed by the `map`-function, that applies a function to every element of a list, producing a new list:

⁸In the language Haskell a single `:` is used for cons, and a double `::` for typing. Type Theoreticians put more emphasis on typing hence they prefer to have the shorter notation for this.

```

map : (A → B) → List A → List B
map f [] = []
map f (x :: xs) = f x :: map f xs

```

So for example

```

map suc (1 :: 2 :: 3 :: []) = 2 :: 3 :: 4 :: []

```

As for natural numbers recursive definitions over lists can be reduced to one combinator, which is usually called `fold` but which we consequently call `ItList`:

```

ItList : M → (A → M → M) → List A → M
ItList n c [] = n
ItList n c (a :: l) = c a (ItList n c l)

```

All the recursive functions over lists we have defined can be reduced to `ItList` using the ideas we have established in the previous section:

```

length-it : List A → ℕ
length-it = ItList 0 (λ a length-l → suc length-l)
sumList-it : List ℕ → ℕ
sumList-it = ItList 0 _+_
_++it_ : {A : Set} → List A → List A → List A
_++it_ = ItList (λ l → l) (λ a m++l → a :: (m++l))
map-it : {A B : Set} → (A → B) → List A → List B
map-it f = ItList [] (λ a map-f-l → f a :: map-f-l)

```

We can also derive a recursor for lists and other inductive datatypes which allows to use the input list in the recursive call but we leave the details as an exercise.

4.3.2 Syntax trees

A common application of inductive datatypes is the representation of syntax. An expression like $2 * (3 + 4)$ can be viewed as a tree generated by constructors for constants and for each of the operators:

```

data Expr : Set where
  const : ℕ → Expr
  _[_+]_ : Expr → Expr → Expr
  _[_*]_ : Expr → Expr → Expr

```

So our example expression can be represented as follows:

```

const 2 [_*] (const 3 [_+] const 4) : Expr

```

An obvious function we want to define is the evaluator which computes the numeric value of an expression:

```

eval : Expr → ℕ
eval (const x) = x
eval (e1 [+] e2) = eval e1 * eval e2
eval (e1 [*] e2) = eval e1 + eval e2

```

So for example

```
eval (const 2 [*] (const 3 [+] const 4)) = 14
```

As for other inductive types we introduce an iterator for Expr:

```

ltExpr : (ℕ → M) → (M → M → M)
        → (M → M → M)
        → Expr → M
ltExpr c p m (const x) = c x
ltExpr c p m (e1 [+] e2) = p (ltExpr c p m e1) (ltExpr c p m e2)
ltExpr c p m (e1 [*] e2) = m (ltExpr c p m e1) (ltExpr c p m e2)

```

I hope that the general scheme for deriving iterators is now evident. The methods correspond to the constructors where every occurrence of the datatype is replaced with the motive M. The function which is computed by the iterator recursively replaces each constructor with the method.

We can define the evaluator using the iterator:

```

eval-it : Expr → ℕ
eval-it = ltExpr
        (λ n → n)
        (λ eval-e1 eval-e2 → eval-e1 + eval-e2)
        (λ eval-e1 eval-e2 → eval-e1 * eval-e2)

```

4.3.3 Rose trees

The syntax trees from the previous section were binary trees, the nodes created by `_[+]_` and `_[*]_` each have two subtrees. Using lists of subtrees we can define trees with an arbitrary number of subtrees which are called *rose trees*.

```

data RoseTree : Set where
  node : List RoseTree → RoseTree

```

We don't need a special constructor for leaves, because we can just have a node with no subtrees `node []`.

As an example we define the following rose tree:

```

rt : RoseTree
rt = node (node []
  :: node (node (node [] :: []) :: node [] :: [])

```

```

:: node (node [] :: [])
:: []

```

As an example for a recursive function on rose trees we count the number of nodes:

```

countNodes : RoseTree → ℕ
countNodes (node ts) = suc (countNodesList ts)
where countNodesList : List RoseTree → ℕ
countNodesList [] = 0
countNodesList (t :: ts) = countNodes t + countNodesList ts

```

Here we use an auxiliary function which counts the nodes in a list of rose trees. We use **where** to do this. ⁹

So for example `countNodes rt` evaluates to 8.

What is the iterator for Rose Trees? We just apply the scheme and obtain the type of methods by replacing `RoseTree` with the motive. We end up with

```

ItRoseTree : ((List M) → M) → RoseTree → M

```

But how can we recursively apply the iterator to the arguments which are inside a list. A good way to address this is to use `map`:

```

ItRoseTree m (node ts) = m (map (ItRoseTree m) ts)

```

However, Agda's termination checker isn't clever enough to see that recursion through `map` will terminate and we have to use the following more elaborate version where we implement an instance of `map` locally to the iterator:

```

ItRoseTree {M} m (node ts) = m (maplt ts)
where maplt : List RoseTree → List M
maplt [] = []
maplt (t :: ts) = ItRoseTree m t :: maplt ts

```

Now we can derive our count function from the iterator. The method is the function `sumList` which, as we have seen, can be defined using only the iterator for lists:

```

countNodes-it : RoseTree → ℕ
countNodes-it = ItRoseTree (λ ns → suc (sumList ns))

```

4.3.4 Ordinals

All the trees we have seen so far were *finitely branching* that is any node only had finitely many subtrees. We can also consider infinitely branching trees and a nice application of those is a notation for ordinal numbers.

⁹Agda has both **let** and **where**, but for reasons I don't understand only **where** supports the definition of functions by pattern matching.

The idea of ordinal numbers is that we also consider infinite numbers. That is we have the sequence

$$0, 1, 2, 3, 4, \dots$$

but now we add an element on the top which we call ω :

$$0, 1, 2, 3, 4, \dots \omega$$

The idea is that ω is greater than any of the numbers in the sequence $0, 1, 2, 3, 4, \dots$ and it is the smallest such ordinal. We can continue this and create a new infinite sequence:

$$0, 1, 2, 3, 4, \dots \omega, \omega + 1, \omega + 2, \omega + 3, \omega + 4, \dots$$

And again we have an element on the top:

$$0, 1, 2, 3, 4, \dots \omega, \omega + 1, \omega + 2, \omega + 3, \omega + 4, \dots \omega + \omega$$

We can continue this game infinitely and produce yet another ordinal $\omega * \omega$ and so on. The main idea is that every ordinal number is associated with the corresponding $_ < _$ relation. In each case this relation is well-founded, that is you can only go down a finite number of steps before reaching 0. Note that this also holds for ω , in the first step we can go down to any finite number and after this it is clear that we will end up at 0. However, there is no bound to the length of this sequence.

Here is the definition of our notation for ordinals. It starts like the definition for natural numbers but then we add another constructor `lim` which allows us to construct ordinals from infinite sequences:

```
data Ord : Set where
  zero : Ord
  suc  : Ord → Ord
  lim  : (ℕ → Ord) → Ord
```

We can embed the natural numbers into the ordinals:

```
emb : ℕ → Ord
emb zero = zero
emb (suc x) = suc (emb x)
```

Using `emb` we can derive ω :

```
ω : Ord
ω = lim emb
```

We can lift the arithmetic operations to ordinals to derive ordinals like $\omega + \omega$ and $\omega * \omega$. Before we define addition it is a good idea to note that ordinal addition is not commutative, that is $\alpha + \beta$ is not the same as $\beta + \alpha$. We can understand addition by just putting the ordinals next to each other. So for example $1 + \omega$ looks like this:

$$*, 0, 1, 2, 3, 4, \dots$$

Here we write $*$ for the 0 in the ordinal 1. On the other hand $\omega + 1$ looks like this

$$0, 1, 2, 3, 4, \dots *$$

In the first case we have just shifted the ordinals by 1, but the order we obtain is equivalent to the one of ω , that is in the sense of orders the ordinals $1 + \omega$ and ω are equivalent. This is not the case for $\omega + 1$ because this time there is an infinite gap between the finite elements and the new one, which is not equivalent to ω .

While when defining $_ + _$ for natural numbers it didn't matter over which argument we did the recursion for the ordinals it does. And the correct choice is to do recursion over the 2nd argument:

$$\begin{aligned} _ +_{\text{ord}} _ &: \text{Ord} \rightarrow \text{Ord} \rightarrow \text{Ord} \\ m +_{\text{ord}} \text{zero} &= m \\ m +_{\text{ord}} \text{suc } n &= \text{suc } (m +_{\text{ord}} n) \\ m +_{\text{ord}} \text{lim } f &= \text{lim } (\lambda i \rightarrow m +_{\text{ord}} f i) \end{aligned}$$

We define addition for a limit ordinal, that is an ordinal of the form $\text{lim } f$ by adding the same element to each element of the sequence and then taking the limit of the sequence. So for example $\omega + \omega$ is obtained by taking the limit of the sequence $\omega, \omega + 1, \omega + 2, \dots$ matching our intuitive explanation.

Ordinals can be used to quantify the recursive strength of a recursion scheme but also the strength of a logic. A famous example is related to the fight of Hercules against the Hydra. The Hydra is a monsters with many heads which we view as a rose tree. Hercules can chop of one head of the Hydra (that is remove a subtree of the form $\text{node } []$) but as a response the Hydra can grow an arbitrary number of heads at lower levels. However, Hercules will always win after a finite number of steps and be able to chop off all the heads.

The fact that Hercules will win is an instance of a particular ordinal called ϵ_0 which we can obtain by defining multiplication and exponentiation and then take the limit of ω -towers of the form

$$\omega, \omega^\omega, \omega^{\omega^\omega}, \dots$$

Any Hydra configuration can be viewed as an ordinal in this sequence and any move by Hercules chooses a smaller ordinal. However, one can show that the ordinal ϵ_0 corresponds to the logical strength of Arithmetic and hence we cannot prove that the game terminate using only induction and natural numbers.

Before we finish let's derive the iterator for **Ord**:

$$\begin{aligned} \text{ItOrd} &: \{M : \text{Set}\} \rightarrow M \rightarrow (M \rightarrow M) \rightarrow ((\mathbb{N} \rightarrow M) \rightarrow M) \\ &\rightarrow \text{Ord} \rightarrow M \\ \text{ItOrd } z \text{ s l } \text{zero} &= z \\ \text{ItOrd } z \text{ s l } (\text{suc } x) &= s (\text{ItOrd } z \text{ s l } x) \\ \text{ItOrd } z \text{ s l } (\text{lim } f) &= l (\lambda i \rightarrow \text{ItOrd } z \text{ s l } (f i)) \end{aligned}$$

and implement ordinal addition as an instance:


```

_+ord-it_ : Ord → Ord → Ord
α +ord-it β = ItOrd α suc lim β

```

Note that we explicitly do recursion over the 2nd argument.

4.4 Positivity

Introductions to λ -calculus often start with the untyped λ -calculus which presents the syntax of the λ -calculus without types. We can try to simulate this idea by defining a strange datatype Λ :

```

data  $\Lambda$  : Set where
  lam : ( $\Lambda \rightarrow \Lambda$ )  $\rightarrow$   $\Lambda$ 

```

Agda won't accept this definition (for good reasons) but for the moment we can switch this off by saying `NO_POSITIVITY_CHECK`. The idea is that any function $\Lambda \rightarrow \Lambda$ gives rise to another untyped λ -term which corresponds to its λ -abstraction. The simplest example is the identity function $\lambda x \rightarrow x$ which is represented as `lam ($\lambda x \rightarrow x$) : Λ` . To be able to form more interesting terms we need to use application. However, this is easy to define using pattern matching:

```

app :  $\Lambda \rightarrow \Lambda \rightarrow \Lambda$ 
app (lam f) x = f x

```

Since every untyped λ -term is produced by an abstraction, we can just use this to obtain a function and hence we have application. We can obtain some interesting terms such as self-apply which is

$$\lambda x \rightarrow x x$$

Hence using Λ we define:

```

self-apply :  $\Lambda$ 
self-apply = lam ( $\lambda x \rightarrow$  app x x)

```

What happens if we apply `self-apply` to itself? This term is called Ω :

$$\Omega = (\lambda x \rightarrow x x) (\lambda x \rightarrow x x)$$

we translate this to

```

 $\Omega$  :  $\Lambda$ 
 $\Omega$  = app self-apply self-apply

```

Now the strange thing about Ω is that when we β -reduce it we get back to Ω :

$$\begin{aligned}
\Omega &= (\lambda x \rightarrow x x) (\lambda x \rightarrow x x) \\
&= (x x) [x \mapsto (\lambda x \rightarrow x x)] \\
&= (\lambda x \rightarrow x x) (\lambda x \rightarrow x x) \\
&= \Omega
\end{aligned}$$

This means that we have constructed an element of Λ which is not generated by `lam`. And indeed if we try to evaluate Ω Agda will hang.

It is the negative occurrence of the type we are defining which causes this. Another example of a type which only has a negative occurrence is the following (and again we have to switch off the positivity checker).

```
data Weird1 : Set where
  foo : (Weird1 → ⊥) → Weird1
```

It is hard to explain `Weird1` but that is the point. We can show that `Weird1` is empty using pattern matching. The idea is that if there is an element of `Weird1` then we have a function `Weird1 → ⊥` and hence we can derive \perp :

```
¬weird1 : Weird1 → ⊥
¬weird1 (foo x) = x (foo x)
```

However, now using `¬weird1` we can actually construct an element of `Weird1`, namely `foo ¬weird1`. But since we have shown that `Weird1` is empty we can derive an element of the empty type:

```
bad : ⊥
bad = ¬weird1 (foo ¬weird1)
```

This is certainly bad, because the empty type was supposed to be empty.

These examples illustrate that we need to have some restrictions on the types of constructors when defining an inductive datatype. The question to answer first is what do we actually mean by an inductive datatype? All the elements of an inductive datatype are generated from the constructors. In the case that an argument of the constructor is the type we are defining means that this should be an element which we know already. That is we create the elements in stages, e.g. with the natural numbers in the first step we can only create `zero` : \mathbb{N} , or for the expressions we can only create the constants `const 1` : `Expr`. But in the next step we can create `suc n` : \mathbb{N} or in the case of `Expr` we can create for example `const 1 [+] const 2` and so on. This also justifies the recursion principles expressed via the iterators. This line of thought explains inductive datatypes where the type appears as an argument to a constructor.

This also works for rose trees but here we have to create a whole list of trees before we can go on creating a new tree.

An interesting example is

```
data T1 : Set where
  foo : T1 → T1
```

What are the elements of this type? Since we have no leaves we can't even get started to create elements. Hence the type is empty. We can actually show this by defining a function into the empty type by recursion:

```
¬t1 : T1 → ⊥
¬t1 (foo x) = ¬t1 x
```

Understanding infinitely branching trees like `Ord` is a bit more involved. We first create all the trees we can build in finitely many stages. These include some of the `lim`-trees but only the one which only use the trees we have constructed before, hence which have only a finite number of different subtrees. Once we have created all of them we have reached a stage ω . Now we can continue to create trees using all the infinitely many we have created up to ω . And so on, we can repeat this process to get more and more trees.

Hence we can allow constructor arguments which are of the form $A \rightarrow X$ where X is the type we are defining just now and A is a type we already know. This is called a *strictly positive type* and this is the condition Agda is using to decide whether a datatype declaration is acceptable.

But what about the following type?

```
data Weird2 : Set where
  foo : ((Weird2 → Bool) → Bool) → Weird2
```

In this example the occurrence of `Weird2` in the argument is not strictly positive but still positive, i.e. double negative.

Should we allow such types? Agda doesn't think so, but why? First of all our semantic explanation is out of the window, we cannot generate `Weird2` in stages. But is it inconsistent? This doesn't seem to be the case even though the proof needs quite strong logical principles. However, this type is inconsistent with classical logic.

If we assume TND or RAA then the type `Bool` is the type of propositions and `Weird2` constructs a fixpoint of the double power set and we can use well known constructions (Cantor's diagonalisation) to derive a contradiction.

Hence, while positive datatypes on their own seem to be consistent, they rule out certain assumptions like the assumption of classical logic. Moreover they are not justified semantically. These seem to be good reasons to rule them out.

4.5 Coinductive types

One of the nice features of category theory is the availability of a mirror: for any construction X there is a dual one, which is usually called `co-X`. For example from the point of category products are dual to sums and indeed sums are sometimes called coproducts. We have seen that recursive sums aka inductive types are very useful, hence now we wonder what can we do with recursive products or coinductive types.

4.5.1 Streams

An example of a coinductive type is the type of streams, or infinite sequences. Given a stream over A we can obtain its `head` which is an element of A and its `tail` which is another stream. Hence we define:

```

record Stream (A : Set) : Set where
  coinductive
  field
    head : A
    tail : Stream A

```

Unlike in the case for sums we have to indicate that we want to define a recursive record by invoking the keyword `coinductive`.

To construct a stream we just have to present a head and a tail. We can derive a constructor for streams using copattern matching which we have already seen for products:

```

_ :: _ : A → Stream A → Stream A
head (a :: as) = a
tail (a :: as) = as

```

Indeed we can automatically derive the constructor by saying:

```

record Stream (A : Set) : Set where
  constructor _ :: _
  coinductive
  field
    head : A
    tail : Stream A

```

We can use copatternmatching which we have already seen for products to define the function

```

from : ℕ → Stream ℕ

```

which for any number n returns the stream of numbers starting with n that is $n, n+1, n+2, \dots$

```

head (from n) = n
tail (from n) = from (suc n)

```

Clearly the head of `from n` is just n , while the tail is the stream starting with the next number, hence `from (suc n)`.

While we understood inductive types by construction, that is all elements are constructed using the constructors, we understand coinductive types by the way they are destructed. E.g. a stream is anything of which we can take a head and a tail.

The recursive definition of `from n` is reasonable, because we can always compute the head and the tail of the result of `from n`, we never get stuck.

Clearly we cannot define `from` for lists

```

fromList : ℕ → List ℕ
fromList n = n :: fromList (suc n)

```

because the recursive definition is not justified by descending along an inductively defined type eg \mathbb{N} .

On the other hand while we can define a function that filters out elements from a list given a boolean function:

```
filterL : {A : Set} → (A → Bool) → List A → List A
filterL f [] = []
filterL f (x :: xs) with f x
filterL f (x :: xs) | false = filterL f xs
filterL f (x :: xs) | true  = x :: filterL f xs
```

However, Agda will complain if we try to do the same for streams:

```
filterS : {A : Set} → (A → Bool) → Stream A → Stream A
filterS f xs with f (head xs)
filterS f xs | false = filterS f (tail xs)
filterS f xs | true  = (head xs) :: filterS f (tail xs)
```

Indeed, we cannot answer the question what is the `head` of `filterS f xs`. And indeed there may be none, we can define a stream of `false`:

```
falses : Stream Bool
head falses = false
tail falses  = falses
```

but what is `head (filterS (λ x → x) falses) : Bool`?

4.5.2 Conatural numbers

Actually streams are not the dual of lists because they do not allow empty streams. We can define a type of colists which contain both finite and infinite sequences but instead we are going to study the dual of natural numbers, the conatural numbers. As we have observed previously the destructor for natural numbers is `pred : $\mathbb{N} \rightarrow \text{Maybe } \mathbb{N}$` . Hence we define conatural numbers as objects that have a predecessor:

```
record  $\mathbb{N}_\infty$  : Set where
  coinductive
  field
    pred $_\infty$  : Maybe  $\mathbb{N}_\infty$ 
```

As we have derived the predecessor for the natural numbers we can now derive `0` and `suc` for conatural numbers:

```
zero $_\infty$  :  $\mathbb{N}_\infty$ 
pred $_\infty$  zero $_\infty$  = nothing
suc $_\infty$  :  $\mathbb{N}_\infty \rightarrow \mathbb{N}_\infty$ 
pred $_\infty$  (suc $_\infty$  n) = just n
```

The rationale behind these definitions should be clear: zero_∞ is the conatural number on which pred_∞ returns an error, while the successor of a conatural number is the one whose predecessor is **just** that number.

Apart from those finite numbers we can also create an infinite number:

$$\begin{aligned} \infty &: \mathbb{N}_\infty \\ \text{pred}_\infty \infty &= \text{just } \infty \end{aligned}$$

The definition is straightforward: what is the predecessor of ∞ ? It is certainly not 0 but it is **just** ∞ .

Let's define addition for conatural numbers. It is a good idea to have the definition of addition of natural numbers in mind:

$$\begin{aligned} _+_&: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{zero} + n &= n \\ \text{suc } m + n &= \text{suc } (m + n) \end{aligned}$$

In contrast to define $_+_\infty$ we have to determine what is the predecessor of $m +_\infty n$?

$$\begin{aligned} _+_\infty &: \mathbb{N}_\infty \rightarrow \mathbb{N}_\infty \rightarrow \mathbb{N}_\infty \\ \text{pred}_\infty (m +_\infty n) &\mathbf{with} \text{ pred}_\infty m \\ \text{pred}_\infty (m +_\infty n) \mid \text{nothing} &= \text{pred}_\infty n \\ \text{pred}_\infty (m +_\infty n) \mid \text{just } m' &= \text{just } (m' +_\infty n) \end{aligned}$$

To find this out we query the first argument. If it was 0, that is $\text{pred}_\infty m$ returns **nothing** then it is the predecessor of the 2nd argument. However, if the predecessor is **just** m' then the result is **just** $m' +_\infty n$.

$_+_\infty$ is the infinite extension of $_+_&$. To make this statement a bit more precise, we can define an embedding

$$\begin{aligned} \mathbb{N} \rightarrow \mathbb{N}_\infty &: \mathbb{N} \rightarrow \mathbb{N}_\infty \\ \mathbb{N} \rightarrow \mathbb{N}_\infty \text{ zero} &= \text{zero}_\infty \\ \mathbb{N} \rightarrow \mathbb{N}_\infty (\text{suc } n) &= \text{suc}_\infty (\mathbb{N} \rightarrow \mathbb{N}_\infty n) \end{aligned}$$

just using recursion. Now the defining property of $_+_\infty$ wrt $_+_&$ is

$$\mathbb{N} \rightarrow \mathbb{N}_\infty (m + n) \equiv (\mathbb{N} \rightarrow_\infty m) +_\infty (\mathbb{N} \rightarrow_\infty n)$$

We haven't yet developed the machinery to prove this and we haven't explained when two coinductive numbers are equal.

4.5.3 Coiterator and corecursor

As for recursion over an inductive type we can derive a general combinator to do corecursion over a coinductive type. To define a function $f : W \rightarrow \text{Stream } A$ corecursively we are using copattern patching:

```

head (f x) = h x
tail (f x) = f (t x)

```

where

```

h : A → W
t : W → W

```

This scheme of *coiteration* can be expressed as one higher order function:

```

ColtStream : {A W : Set} → (W → A) → (W → W)
              → W → Stream A
head (ColtStream h t x) = h x
tail (ColtStream h t x) = ColtStream h t (t x)

```

As an example the from function can be defined using only the coiterator:

```

from-coit : ℕ → Stream ℕ
from-coit = ColtStream (λ n → n) (λ n → suc n)

```

Similarly in the case of the conatural numbers we can define:

```

Coltℕ∞ : {W : Set} → (W → Maybe W) → W → ℕ∞
pred∞ (Coltℕ∞ p x) with p x
pred∞ (Coltℕ∞ p x) | nothing = nothing
pred∞ (Coltℕ∞ p x) | just x' = just (Coltℕ∞ p x')

```

As an example we can derive

```

∞-coit : ℕ∞
∞-coit = Coltℕ∞ just tt

```

However, it turns deriving `_+∞_` is a bit more difficult. Here is the code we have written using `copatternmatching`:

```

_+∞_ : ℕ∞ → ℕ∞ → ℕ∞
pred∞ (m +∞ n) with pred∞ m
pred∞ (m +∞ n) | nothing = pred∞ n
pred∞ (m +∞ n) | just m' = just (m' +∞ n)

```

We clearly do corecursion on the first argument. However, if the first argument is 0, that is of `pred∞ m` is `nothing` we immediately return `pred∞ n`. This sort of behaviour cannot be directly achieved with the coiterator, but requires a *corecursion*. The corecursor can stop corecursion and just return an answer, hence the method returns a sum:

```

CoRℕ∞ : {W : Set} → (W → ℕ∞ ⊕ Maybe W) → W → ℕ∞
pred∞ (CoRℕ∞ p x) with p x
pred∞ (CoRℕ∞ p x) | inj1 n = pred∞ n

```

```

pred $\infty$  (CoRN $\infty$  p x) | inj2 nothing = nothing
pred $\infty$  (CoRN $\infty$  p x) | inj2 (just x') = just (CoRN $\infty$  p x')

```

Using the corecursor we can define $_ +_{\infty} _$. We are also using a case combinator for Maybe

```

caseMaybe : {A M : Set} → M → (A → M) → Maybe A → M
caseMaybe n j nothing = n
caseMaybe n j (just x) = j x

```

```

 $\_ +_{\infty}$ -corec $\_ : \mathbb{N}_{\infty} \rightarrow \mathbb{N}_{\infty} \rightarrow \mathbb{N}_{\infty}$ 
m  $\_ +_{\infty}$ -corec n =
  CoRN $\infty$  ( $\lambda x \rightarrow$  caseMaybe (inj1 n) ( $\lambda m' \rightarrow$  inj2 (just m')) (pred $\infty$  x)) m

```

The corecursor is dual to the recursor, which can be also formulated using a product but this is curried in the usual formulation. And as for the recursor, the corecursor can be derived from the coiterator.

4.6 Functorial semantics

Ok, we have seen a number of examples of inductive and coinductive definitions but what is the general scheme. Let's start with inductive definitions, for example once again the natural numbers:

```

data  $\mathbb{N} : \text{Set}$  where
  zero :  $\mathbb{N}$ 
  suc  :  $\mathbb{N} \rightarrow \mathbb{N}$ 

```

Instead of using two constructors we could have used a sum and combine them into one:

```

data  $\mathbb{N}' : \text{Set}$  where
  zerosuc : Maybe  $\mathbb{N}' \rightarrow \mathbb{N}'$ 

```

The function zerosuc which we defined previously now is the constructor. We recover zero and suc as

```

zero' :  $\mathbb{N}'$ 
zero' = zerosuc nothing
suc'  :  $\mathbb{N}' \rightarrow \mathbb{N}'$ 
suc' n = zerosuc (just n)

```

What about the iterator? Its type should be the following:

```

lt $\mathbb{N}' : (\text{Maybe } M \rightarrow M) \rightarrow \mathbb{N}' \rightarrow M$ 

```

Now we want to define lt \mathbb{N}' m (zerosuc n) by recursively using m and lt \mathbb{N}' on n. However, the problem is that $n : \text{Maybe } \mathbb{N}$ so we cannot directly apply it.

However, luckily `Maybe` is a *functor* that is it comes with a `map` function so that we can apply it to functions:

```
map-Maybe : (A → B) → Maybe A → Maybe B
map-Maybe f nothing = nothing
map-Maybe f (just x) = just (f x)
```

Now using `map-Maybe` we can complete the definition of `lt-ℕ'`:¹⁰

```
lt-ℕ' m (zerosuc n) = m (map-Maybe (lt-ℕ' m) n)
```

This can be turned into a general pattern: an inductive datatype is given by a functor (often called the signature functor), i.e. an operation

```
T : Set → Set
```

together with a `map` function:

```
map-T : (A → B) → T A → T B
```

The datatype definition μT then is given by just one constructor:


```
data  $\mu T$  : Set where
  in-T : T  $\mu T$  →  $\mu T$ 
```

and the iterator is:

```
lt-T : (T M → M) →  $\mu T$  → M
lt-T m (in-T x) = m (map-T (lt-T m) x)
```

This is a good opportunity to draw a categorical diagram:

¹⁰Actually Agda rejects this definition because the recursion goes through `map-Maybe` which is defined separately. We could have fixed this by defining `map-Maybe-lt-ℕ'` mutually but here I just said `{-# TERMINATING #-}`.



This diagram expresses that

$$\text{in-T} \circ (\text{mapT (lt-M m)}) = (\text{lt m}) \circ m$$

In general categorical diagrams express that different paths through the diagram yield the same result, where sequencing of arrows is interpreted as composition.

We can also describe the recursor in this framework:

$$\begin{aligned}
 \text{R-T} &: (T (\mu T \times M) \rightarrow M) \rightarrow \mu T \rightarrow M \\
 \text{R-T m (in-T x)} &= m (\text{map-T } (\lambda y \rightarrow y, \text{R-T m y}) x)
 \end{aligned}$$

And indeed we can generically derive it from the iterator:

$$\begin{aligned}
 \text{R-T-it} &: (T (\mu T \times M) \rightarrow M) \rightarrow \mu T \rightarrow M \\
 \text{R-T-it } \{M\} f x &= \text{proj}_2 (\text{lt-T } \{\mu T \times M\} (\lambda p \rightarrow \text{in-T (map-T proj}_1 p), f p) x)
 \end{aligned}$$

We can define functors for all the examples we have considered so far:

$$\begin{aligned}
 \text{T-List} &: \text{Set} \rightarrow \text{Set} \rightarrow \text{Set} \\
 \text{T-List } C X &= \text{Maybe } (C \times X) \\
 \text{T-Expr} &: \text{Set} \rightarrow \text{Set} \\
 \text{T-Expr } X &= \mathbb{N} \uplus X \times X \uplus X \times X \\
 \text{T-RoseTree} &: \text{Set} \rightarrow \text{Set} \\
 \text{T-RoseTree } X &= \text{List } X \\
 \text{T-Ord} &: \text{Set} \rightarrow \text{Set} \\
 \text{T-Ord } X &= \text{Maybe } (X \uplus (\mathbb{N} \rightarrow X))
 \end{aligned}$$

As already indicated we can view codatatypes as the mirror image. Co-datatypes are also given by functors, in the case of streams this is $\text{T-stream } X =$


$A \times X$ and the conatural numbers use `Maybe`, ie. the same as the natural numbers. The coinductive type corresponding to T is:

```
record  $\nu T$  : Set where
  coinductive
  field
    out-T :  $T \nu T$ 
```

and the coiterator is given by copattern matching:

```
Colt-T : (M  $\rightarrow$  T M)  $\rightarrow$  M  $\rightarrow$   $\nu T$ 
out-T (Colt-T f x) = map-T (Colt-T f) (f x)
```

This calls for another diagram:



Comparing this with the previous diagram we can see that the directions of all the arrows have been reversed. I leave it as an exercise to define the corecursor and derive it from the Coiterator.

You may have noticed that I have cheated in a number of ways. None of the definitions I have given are acceptable in Agda as they stand. The definitions of the μT and νT are rejected because they go through T and Agda doesn't know anything about T . Ok, we have assumed that it is a functor but actually this is not enough because `Wierd2` is given by a functor:

```
T-Weird2 : Set  $\rightarrow$  Set
T-Weird2 X = (X  $\rightarrow$  Bool)  $\rightarrow$  Bool
T-Wierd2-map : (A  $\rightarrow$  B)  $\rightarrow$  T-Weird2 A  $\rightarrow$  T-Weird2 B
T-Wierd2-map f g h = g (h  $\circ$  f)
```

We will get back to this later and give a definition which only applies to strictly positive types but we need dependent types for this.

I would also like to add that it is not enough to have `map-T` but to be a proper functor also needs to preserve identity and composition, i.e. `map-T id = id` and `map-T (f ∘ g) = map-T f ∘ map-T g` but we need equality types to specify this in Type Theory.

4.7 History

Giuseppe Peano lived in Turino in the late 19th century. He wrote down the rules of reasoning for the natural numbers which is now in his honour called *Peano Arithmetic*. A proper account of Peano Arithmetic uses predicate logic which we haven't yet introduced. Peano only used addition and multiplication in his presentation but it can be shown that all primitive recursive functions can be derived. Interestingly, this ceases to be the case if you leave out multiplication which leads to something called Pressburger Arithmetic.

By primitive recursive functions one means traditionally functions which can be derived using no function types, that is with `RN` but only products and `N` can be used. This intuitively corresponds to for-loops in programming. Ackermann formulated his function to show that there are computable functions which cannot be defined with primitive recursion.

It was Gödel who formulated a system which can express all functions that are definable in Peano arithmetic. This is called System T ¹¹ and it is basically what we have presented, i.e. you have `RN` and function types. There are still functions which are not definable in this system, corresponding to the problem of Archilles and the hydra which we have mentioned already in the section on ordinals.

Our presentation of inductive types goes back to algebraic datatypes and pattern matching invented by my PhD supervisor Rod Burstall which were first implemented in a language called Hope [BMS80]. Subsequently algebraic datatypes and pattern matching was used in many functional programming languages including Miranda and Haskell. Coinductive types were investigated by Coquand [Coq93] who also formulated the guarded corecursion principle we are using. Copatternmatching was introduced much later in the context of Type Theory by Andreas Abel and Brigitte Pientka [AP13]. The functorial understanding of datatypes and codatatypes can be found in the thesis of Tatsuyo Hagino [Hag87]. ¹²

We only briefly touched upon the untyped λ -calculus which, together with untyped combinatory logic, is a basic computational mechanism indeed an alternative to Turing machines. A good overview over the untyped λ -calculus is Barendregt's book [B⁺84].

¹¹Gödel introduced System T in [Göd58] (in german).

¹²Hagino was also supervised by Rod Burstall, but he finished before I arrived in Edinburgh.

4.8 Exercises

1. Define the following functions using pattern matching and structural recursion on the natural numbers.

- (a) Define a function `even?` that determines whether its input is even.

`even? : ℕ → Bool`

Examples :

`even? 3` `= false`
`even? 6` `= true`

- (b) Define a function `sum` that sums the numbers from 0 until `n-1`.

`sum : ℕ → ℕ`

Examples:

`sum 2` `= 3`
`sum 10` `= 55`

- (c) Define a function `max` that calculates the maximum of 2 numbers.

`max : ℕ → ℕ → ℕ`

Examples:

`max 2 3` `= 3`
`max 10 1` `= 10`

- (d) Define a function `fib` which calculates the n th item of the Fibonacci sequence: 1, 1, 2, 3, 5, 8, 13, ... (each number is the sum of the two previous ones).

`fib : ℕ → ℕ`

Examples:

`fib 0` `= 1`
`fib 6` `= 13`

- (e) Define a function `eq?` that determines whether two numbers are equal.

`eq? : ℕ → ℕ → Bool`

Examples:

`eq? 4 3` `= false`
`eq? 6 6` `= true`

- (f) Define a function `rem` such that `rem m n` returns the remainder when dividing `m` by `suc n` (this way we avoid division by 0).

$$\text{rem} : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$$

Examples:

$$\begin{aligned} \text{rem } 5 \ 1 &= 1 \\ \text{rem } 11 \ 2 &= 2 \end{aligned}$$

- (g) Define a function `div` such that `div m n` returns the result of dividing `m` by `suc n` (ignoring the remainder).

$$\text{div} : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$$

Examples:

$$\begin{aligned} \text{div } 5 \ 1 &= 2 \\ \text{div } 11 \ 2 &= 3 \end{aligned}$$

2. Define all the functions of the previous question but this time only use the iterator `ltN` and/or the recursor `RN`.

Use the following naming convention: if the function was called `f` then call it `f-i` if you only use the iterator, but `f-r` if you use the recursor (and possibly the iterator).

3. In this exercise we implement `tree-sort` to sort a list of natural numbers.

We define the type of trees with labels on the nodes.

```
data Tree (A : Set) : Set where
  leaf : Tree A
  node : Tree A → A → Tree A → Tree A
```

Here is an example tree:

```
node (node leaf 1 (node leaf 2 leaf)) 5 leaf
```

Define a function `tree2list` which collects all the leaves from left to right in a list.

```
tree2list : {A : Set} → Tree A → List A
```

Here is a simple unit test

```
tree2list t = 1 :: 2 :: 5 :: []
```

We are going to produce sorted trees. A sorted tree is one where for every node the leaves on the left have smaller values and the leaves on the right have larger values.

I give you a comparison function on \mathbb{N} .

```

_ ≤? _ : ℕ → ℕ → Bool
zero ≤? m = true
suc n ≤? zero = false
suc n ≤? suc m = n ≤? m

```

Write a function that transforms a list of natural number into a sorted tree.

```
list2tree : List ℕ → Tree ℕ
```

Here is a test:

```

list2tree (10 :: 2 :: 1 :: 5 :: [])
= node (node leaf 1 (node leaf 2 leaf)) 5 (node leaf 10 leaf)

```

Hint: it may be a good idea to write a function

```
insert : ℕ → Tree ℕ → Tree ℕ
```

that inserts one value into a sorted tree maintaining its sortedness.

Using the previous function it is now easy to define `tree-sort` a function that sorts lists.

```
tree-sort : List ℕ → List ℕ
```

Here is a unit test for `treesort`.

```
tree-sort (10 :: 2 :: 1 :: 5 :: []) = 1 :: 2 :: 5 :: 10 :: []
```

4. Reimplement `tree-sort` using only the iterators for lists and trees (also called folds).

The functions should **not** use pattern matching on lists or trees or any recursion but only use the combinators defined here:

```

foldList : {A : Set} {M : Set} →
  M → (A → M → M) → List A → M
foldList mnil mcons [] = mnil
foldList mnil mcons (x :: xs) = mcons x (foldList mnil mcons xs)
foldTree : {A : Set} {M : Set} →
  M → (M → A → M → M) → Tree A → M
foldTree mleaf mnode leaf = mleaf

```

```
foldTree mleaf mnode (node l x r) =
  mnode (foldTree mleaf mnode l) x (foldTree mleaf mnode r)
```

If the original function is called x then the function with folds only is called x -f.

5. We have already defined addition for conatural numbers:

```
_+∞_ : N∞ → N∞ → N∞
prd∞ (m +∞ n) with prd∞ m
prd∞ (m +∞ n) | just m' = just (m' +∞ n)
prd∞ (m +∞ n) | nothing = prd∞ n
```

Our goal is now to define multiplication:

```
_*∞_ : N∞ → N∞ → N∞
```

This is not so easy because we need to make sure that the recursion is syntactically guarded, i.e. the recursive call has to appear directly at the right hand side not inside another function.

Hint: Define an auxilliary function first which solves a more general problem.

Now here is a simple unit test:

$$\mathbb{N}_{\infty} \rightarrow \mathbb{N} (\mathbb{N} \rightarrow \mathbb{N}_{\infty} 3 *_{\infty} \mathbb{N} \rightarrow \mathbb{N}_{\infty} 5) = 15$$

6. Define the map operations for all the examples of functors:

```
T-List-map : (A → B) → T-List C A → T-List C B
T-Expr-map : (A → B) → T-Expr A → T-Expr B
T-RoseTree-map : (A → B) → T-RoseTree A → T-RoseTree B
T-Ord-map : (A → B) → T-Ord A → T-Ord B
```

7. Dualize the derivation of the generic recursor in section 4.6

That is define the corecursor:

```
CoR-T : (M → T (νT ⊔ M)) → M → νT
```

using copattern matching, `map-T` and guarded recursion.

The show that it can be derived directly from `Colt` without recursion:

```
CoR-T-coit : (M → T (νT ⊔ M)) → M → νT
```